TALLER DE SECUENCIAS CON GEOGEBRA

José Muñoz Santonja, Instituto de GeoGebra de Andalucía
Mariano Real Pérez, Instituto de GeoGebra de Extremadura

¿Qué es una lista en GeoGebra?

En GeoGebra, usando llaves pueden crearse listas que incluyan varios objetos, como puntos, segmentos, circunferencias... y aquellos elementos que se quieran. Algunos ejemplo serían:

G = {A, B, C} establece una lista de tres elementos definidos previamente **A**, **B**, y **C**.

Un ejemplo práctico sería:
L = {(1, 2, 0), (3, 7, 8), (3, 4, 5)} produce una lista de los puntos ingresados sin nominar.

Además, GeoGebra nos permite seleccionar un elemento concreto de una lista previamente creada.

El **comando Secuencia** tiene el siguiente formato:

Secuencia[<Expresión>, <Variable>, <Valor Inicial>, <Valor Final>, <Incremento>]

Siendo:

<Expresión> Una expresión, fórmula, etc.

<Variable> Es una variable que aparece en la expresión.

<Valor Inicial> y **<Valor Final>** es el intervalo en el que se moverá la variable, siendo <Valor Inicial> el primer valor que tomará.

<Incremento> es cómo se irá incrementando la variable en el intervalo anterior.

A lo largo de las siguientes actividades observaremos cómo podemos utilizar este comando en nuestras construcciones con GeoGebra.

<table>
<tr><td colspan="2">ACTIVIDAD NÚMERO 1</td></tr>
<tr><td>Título de la actividad:</td><td>Construcción de una parábola</td></tr>
<tr><td colspan="2">Enunciado de la actividad</td></tr>
<tr><td colspan="2">Queremos dibujar varios puntos de la parábola $y=x^2$ en el intervalo [0,10]

</td></tr>
<tr><td colspan="2">Pasos que se seguirán con GeoGebra</td></tr>
<tr><td colspan="2">En este caso, en la ventana de entrada de GeoGebra basta escribir:

Secuencia((i, i^2), i, 0, 10, 0.5)</td></tr>
<tr><td colspan="2">Posibles ampliaciones</td></tr>
<tr><td colspan="2">Algunas posibles ampliaciones son:
.- Se pueden dibujar varios puntos de una gráfica cualquiera.
.- Podemos colocar deslizadores para indicar el intervalo en el que se moverá la variable
.- Además, podemos colocar un deslizador para indicar el incremento.

Guarda el archivo generado con el nombre parabola.ggb</td></tr>
</table>

Como hemos visto en la actividad anterior, una lista puede estar compuesta no solamente por números, sino por todo tipo de objetos que permita GeoGebra. En concreto, puede estar constituida por palabras o frases. Veamos una actividad muy simple para comprobar cómo podemos seleccionar un elemento de una lista. Para ello se utiliza la orden:

Elemento(〈lista〉,〈número(posición)〉)

Con ella se toma el elemento que está en la posición indicada por el número, dentro de la lista nombrada al principio.

ACTIVIDAD NÚMERO 2

Título de la actividad:	Construcción aleatoria de polígonos regulares

Enunciado de la actividad

Queremos realizar una construcción, en la que se dibujará aleatoriamente un polígono regular y se complementará con el nombre de ese polígono.

Pasos que se seguirán con GeoGebra

Los pasos a seguir en la construcción son los siguientes:

1.- Primero se crea una lista con los nombres de los polígonos desde tres lados hasta 10.

nombres = {"TRIÁNGULO", "CUADRADO", "PENTÁGONO", "HEXÁGONO", "HEPTÁGONO", "OCTÓGONO", "ENEÁGONO", "DECÁGONO"}

2.- Se define ahora un número que se generará aleatoriamente entre 3 y 10.

nl=AleatorioEntre(3, 10)

3.- Se dibujan dos puntos cualesquiera A y B en la ventana gráfica y se dibuja un polígono regular sobre ellos con la orden:

Polígono(nombres,nl-2)

4.- Para colocar el nombre del polígono usamos la orden:

Texto(Elemento(nombres, nl-2))

5.- Para hacer que el polígono cambie aleatoriamente, podemos añadir un botón incluyendo, en *guión (script)* el comando: ***ActualizaConstrucción()***

Guarda el archivo generado con el nombre **polígonos_regulares.ggb**

Posibles ampliaciones

Se puede ampliar la actividad, quitando que aparezca el nombre y añadiendo una casilla de entrada en la que el alumno debe escribir el nombre que tiene el polígono que le ha aparecido. Tras incluir el nombre le aparece un mensaje si ha sido correcto o no.

La versión anterior sería para Primaria, pero se puede hacer una igual para secundaria creando una lista con funciones y pidiendo que el alumno reconozca el tipo de función que puede ser: afín, lineal, cuadrática, hiperbólica, exponencial, etc.

Guarda el archivo generado con el nombre **polígonos_regulares2.ggb**

ACTIVIDAD NÚMERO 3

Título de la actividad:	Generar una sucesión

Enunciado de la actividad

Queremos generar los primeros términos de una sucesión aleatoria cuyo término general tenga la forma x^y con x e y número enteros, estando y entre 2 y 8



Pasos que se seguirán con GeoGebra

Los pasos a seguir en la construcción son los siguientes:

1.- Se define ahora un número que se generará aleatoriamente entre 1 y 20.

n=AleatorioEntre(1, 15)

2.- Se define ahora un número que se generará aleatoriamente entre 3 y 9.

nterminos=AleatorioEntre(3, 9)

3.- Se define ahora un número que se generará aleatoriamente entre 2 y 8.

exponente=AleatorioEntre(2, 8)

4.- Se introduce la secuencia:

Secuencia((n+i)^exponente, i, 0, nterminos, 1)

5.- Ahora se coloca el texto: “Calcula el término general de la siguiente sucesión:”

6.- Posteriormente se introduce un nuevo texto en el que solamente aparezca la lista generada.

7.- Para hacer que se generen aleatoriamente nuevas sucesiones, podemos añadir un botón incluyendo, en *guión (script)* el comando: ***ActualizaConstrucción()***

Guarda la construcción con el nombre **sucesión.ggb**

ACTIVIDAD NÚMERO 4	
Título de la actividad:	Dividir un segmento en n partes iguales

Enunciado de la actividad

Para esta actividad vamos a generar una construcción en la que observemos la división de un segmento cualquiera en tantas partes como deseemos, indicando estas partes a través de un deslizador.

Pasos que se seguirán con GeoGebra

Los pasos a seguir en la construcción son los siguientes:

.- Genera un deslizador n que tome valores enteros entre 1 y 100.

.- Coloca dos puntos A y B.

.- Genera el vector **u=Vector(A, B)**

.- Genera el segmento AB

.- Indicamos que no se visualice el vector u

.- En la barra de entradas introducimos el comando **Secuencia(A+i/n*u, i, 0, n, 1)**

.- Ahora podemos hacer variar el valor de n utilizando el deslizador y observamos como divide al segmento en n partes iguales.

Guarda la actividad con el nombre **segmento.ggb**

Para la siguiente actividad vamos a utilizar la orden Rota. El funcionamiento de esta orden es el siguiente:

Rota(A,ang,O)

Con ella se rota el punto **A** un ángulo **ang**, respecto a un punto **O**.

ACTIVIDAD NÚMERO 5

Título de la actividad:	Aproximación de polígono a circunferencia (PROFUNDIZACIÓN)

Enunciado de la actividad

Comprueba, utilizando GeoGebra, que si inscribimos un polígono regular en una circunferencia se cumple que:

.- Si aumentamos el número de lados del polígono, el área del mismo se aproxima al área del círculo determinado por la circunferencia.

.- Si aumentamos el número de lados del polígono, el perímetro del mismo se aproxima al perímetro de la circunferencia.

En la construcción realizada debe aparecer la circunferencia y el polígono inscrito en cada momento, así como los valores del área y perímetro de la circunferencia y el polígono en cada momento.



NOTA: Recuerda realizar la circunferencia de radio variable para poder utilizar distintas circunferencias.

Pasos que se seguirán con GeoGebra

Para realizar esta construcción con GeoGebra te proponemos seguir los siguientes pasos:

1.- Crea un deslizador **r** que pueda tomar cualquier valor positivo menor que. Será el radio de la circunferencia.

2.- Crea un deslizador **n** que tome un valor entero entre 3 y 200. Será el número de lados del polígono.

3.- Por comodidad, vamos a colocar como centro de la circunferencia el punto **O=(0,0)**.

4.- Trazamos la circunferencia **c** de centro **O** y radio **r**

5.- Seleccionamos un punto **A** cualquiera que esté en la circunferencia.

6.- En la ventana de entrada introducimos el comando **V=Secuencia(Rota(A,i*2*Pi/n,O),i,0,n,1)** que marcará en la circunferencia los vértices de nuestro polígono.

7.- Trazamos el polígono introduciendo el siguiente comando en GeoGebra: **P=Polígono(V)**

8.- Calculamos el área del círculo con el siguiente comando **a=Pi*r^2**

9.- Calculamos la longitud de la circunferencia **L=Perímetro(c)**

10.- Calculamos el perímetro del polígono: **LP=Perímetro(P)**

11.- Introducimos los textos correspondientes en GeoGebra para observar el valor de los cuatro datos.

Guarda la actividad con el nombre **poligono_inscrito.ggb**

Como hemos visto en la actividad anterior, podemos generar una lista utilizando dentro otra orden. En aquél caso la orden Rota. Ahora la vamos a utilizar también en la siguiente actividad.
Para la misma también vamos a utilizar otra orden que podemos aplicar sobre una lista:

Longitud(⟨lista⟩)

Con ella obtenemos el número de elementos que hay en la lista.

ACTIVIDAD NÚMERO 6

Título de la actividad:	Generación de una Cardioide como envolvente

Enunciado de la actividad

Generar la curva cardioide como envolvente de las circunferencias que tienen centro en una misma circunferencia base y que pasan por un punto común de esa circunferencia base.



Pasos que se seguirán con GeoGebra

Para realizar esta construcción con GeoGebra te proponemos seguir los siguientes pasos:

1.- Crea un deslizador **r** que pueda tomar cualquier valor positivo menor que 10. Será el radio de la circunferencia base.

2.- Crea un deslizador **n** que tome un valor entero entre 3 y 200. Será el número de elementos que determinarán la envolvente.

3.- Por comodidad, vamos a colocar como centro de la circunferencia el punto **O=(0,0).**

4.- Trazamos la circunferencia c de centro **O** y radio **r**

5.- Seleccionamos un punto A cualquiera que esté en la circunferencia.

6.- En la ventana de entrada introducimos el comando **V=Secuencia(Rota(A,i*2*Pi/n,O),i,0,n,1)** que marcará en la circunferencia los vértices de nuestro polígono.

7.- Trazamos la circunferencia que tienen como centro cada uno de los puntos de la secuencia anterior y que pasa por el punto A**:**
trazacircunferencia=Secuencia(Circunferencia(Elemento(V, k), A), k, 1, Longitud(V))

Guarda la construcción con el nombre **cardioide.ggb**

ACTIVIDAD NÚMERO 7 (PROFUNDIZACIÓN)

Título de la actividad:	Generación de una parábola como envolvente

Enunciado de la actividad

Obtener una parábola como envolvente de sus tangentes. Para ello, utilizar la definición de parábola como el lugar geométrico de los puntos que equidistan de una recta y de un punto fijo llamado foco.



Pasos que se seguirán con GeoGebra

Para realizar esta construcción con GeoGebra te proponemos seguir los siguientes pasos:

1.- Crea un deslizador **r** que pueda tomar cualquier valor positivo **menor que 20**. Será la desviación que pongamos de los puntos en la recta respecto a la vertical desde el foco.

2.- Crea un deslizador **n** que tome un valor entero **entre 3 y 500**. Será el número de elementos que determinarán la envolvente.

3.- Por comodidad, vamos a colocar como recta **t el eje de abcisas**.

4- Por comodidad, vamos a colocar el punto **F** (foco) como un punto del eje de ordenadas.

5.- En la ventana de entrada introducimos el comando **P=Secuencia((-r,0)+2*r*i/n*(1,0), i, 0, n)** que marcará los n puntos en el intervalo **(-r,r)**.

6.- En la ventana de entrada introducimos el comando **M=Secuencia(PuntoMedio(F, Elemento(P, k)), k, 1, n + 1)** que marcará los puntos medios entre el foco y cada punto de la recta **t**.

7.- En la ventana de entrada introducimos el comando **h=Secuencia(Segmento(F, Elemento(P, z)), z, 1, n + 1)** que marcará el segmento que une el foco con cada punto de la recta t.

8.- En la ventana de entrada introducimos el comando **s=Secuencia(Perpendicular(Elemento(M, w), Elemento(h, w)), w, 1, n + 1)** que marcará cada una de las rectas que buscamos.

9.- Dejamos no visible las listas **M, P y h**

Guarda la construcción como **envolvente.ggb**

ACTIVIDAD NÚMERO 8

Título de la actividad:	Ejemplo de orden Ejecuta

Enunciado de la actividad

Crea una construcción de GeoGebra en la que aparezcan tres puntos A, B y C.

Posteriormente, introduce la lista: L = {"A=(1,1)", "B=(3,3)", "C=Midpoint[A,B]"}

Crea un botón con la imagen que aparece a la derecha u otra parecida, de forma que al pulsarlo se ejecute L.

Comprueba el efecto causado.

Pasos que se seguirán con GeoGebra

Para realizar esta construcción con GeoGebra te proponemos seguir los siguientes pasos:

1.- En la ventana de entrada escribimos **L = {"A=(1,1)", "B=(3,3)", "C=Midpoint[A,B]"}**

2.- Colocamos la imagen en la pantalla de GeoGebra.

3.- Aparecen dos puntos sobre los que se apoya. Se llaman A y B

4.- Vamos a las propiedades del botón y en el "Programa de guión (scripting)", en la parte de "Al hacer clic" escribimos **Ejecuta(L)**

5.- Introduce el texto "**Pulsa sobre el botón**"

6.- Para finalizar, pulsamos sobre el botón y observamos el efecto

Guarda la construcción con el nombre **ejecuta.ggb**

ACTIVIDAD NÚMERO 9

Título de la actividad:	Sucesión de Fibonacci con secuencias dependientes

Enunciado de la actividad

Con la misma orden Ejecuta, vamos a poder crear secuencias dependientes. Un ejemplo claro sería la generación de la sucesión de Fibonacci. En este caso nos proponemos obtener los k primeros términos de la sucesión de Fibonacci.

Pasos que se seguirán con GeoGebra

Para realizar esta construcción con GeoGebra te proponemos seguir los siguientes pasos:

1.- Crea un deslizador **k** que tome valores enteros entre **1 y 30**

2.- En la ventana de GeoGebra crea un botón y denomínalo "**Ejecuta**".

3.- Vamos a las propiedades del botón y en el "Programa de guión (scripting)", en la parte de "Al hacer clic" escribimos
Ejecuta[Secuencia["f_{"+g+"}=0",g,1,32]]

Ejecuta[Encadena[{"f_{1}=1","f_{2}=1"},
Secuencia["f_{"+(n+2)+"}=f_{"+(n+1)+"}+f_{"+n+"}", n, 1, k]]]

4.- Para finalizar, seleccionamos el valor k que consideremos, pulsamos sobre el botón y observamos el efecto.

Guarda la construcción con el nombre **fibonacci.ggb.**

En la siguiente actividad vamos a tener que elegir elementos, de una forma aleatoria, de una lista formada por listas. Hay dos métodos para conseguir seleccionar esos elementos:

Uno sería elegir un elemento aleatorio dentro de la lista, que saldría una lista en sí, y dentro de ella elegir un elemento aleatorio. Esto se haría con la orden:

ElementoAleatorio(ElementoAleatorio(lista))

Otra opción es utilizar la orden

Elemento(lista, índice 1, índice 2)

Esta orden funciona, eligiendo dentro las listas que forman la lista original, según el valor indicado en el primer índice, y dentro de ella, el elemento correspondiente al segundo índice. Esta opción es la que hemos utilizado en la construcción.

ACTIVIDAD NÚMERO 10

Título de la actividad:	Polígonos aleatorios en una trama

Enunciado de la actividad

Vamos a construir una trama de puntos, con la orden secuencia, y generar aleatoriamente cinco de ellos para que el alumno dibuje polígonos sobre esos puntos.



Pasos que se seguirán con GeoGebra

Para construir la trama basta utilizar una secuencia doble. Utilizamos la orden:

puntos=Secuencia(Secuencia((p,q), p, 0, 10), q, 0, 10)

Esta orden nos da una lista formada por 11 listas, cada una de ellas con 11 puntos. Tenemos que elegir, aleatoriamente, cinco puntos. Para ello, tenemos que elegir aleatoriamente la lista de las 11 posibles, y dentro de ella el punto de los 11 disponibles. Utilizamos la siguiente orden:

vértices=Secuencia(Elemento(puntos,AleatorioEntre(1, 11),AleatorioEntre(1, 11)),i,1,5)

Que repite cinco veces el elegir una lista de las que forman puntos y dentro de ella un punto.

A continuación podemos añadir dos botones, uno que elimine el polígono que va a dibujar el alumno. Basta añadir como guiones la orden Elimina[A] y hacer lo mismo para los cinco puntos que se forman al dibujar el polígono, es decir hasta Elimina[E].

Podemos construir otro botón con las mismas órdenes anteriores y además la de actualizar construcción, para que genere un nuevo problema.

Actividades a realizar por el alumno

Una vez que tengamos la trama, el alumno debe dibujar un polígono que una los puntos y contestar a una serie de preguntas como las siguientes:

- ¿Es un polígono convexo o cóncavo?
- ¿Cuántos lados tiene? (*la aleatoriedad puede hacer que se repita algún punto o que salgan puntos alineados, con lo que se pierden lados)*.
- Calcula el área del polígono dibujado.
- ¿Es posible hacer varios polígonos? ¿Cuál sería el de mayor área?
- ¿Cuál es el perímetro del polígono?

Guarda la construcción con el nombre **polígono_aleatorio.ggb.**

ACTIVIDAD NÚMERO 11	
Título de la actividad:	Desarrollo del Binomio de Newton
Enunciado de la actividad	
Vamos a utilizar las listas para conseguir el desarrollo del binomio $\left(x^2+x\right)^3$. Para ello vamos a crear los elementos correspondientes a su desarrollo, es decir los monomios de la forma $\binom{3}{t}\left(x^2\right)^{3-t}\cdot x^t$ y posteriormente sumarlos.	
Pasos que se seguirán con GeoGebra	
Basta escribir la orden: ***términos=Secuencia(NúmeroCombinatorio(3, t)*(x^2)^(3-t)*x^t, t, 0, 3)*** Una vez obtenidos los sumandos del desarrollo, basta sumarlos con la orden, previamente es conveniente ocultar las gráficas dibujadas: ***f(x)=Suma(términos)*** Como los términos quedan un poco expandidos, lo mejor es efectuar las operaciones correspondientes, para ello basta simplificar la expresión obtenida. ***Simplifica(f(x))*** Guarda la construcción con el nombre **binomio_newton.ggb.**	

ACTIVIDAD NÚMERO 12	
Título de la actividad:	Raíces de un número complejo

Enunciado de la actividad

Calcula la raíz n-ésima de un complejo que introduzcas en la ventana de GeoGebra.

Guarda el archivo generado con el nombre **complejo.ggb**



Pasos que se seguirán con GeoGebra

Pasos en la construcción:
Para hacerlo, sigue los siguientes puntos paso a paso:

1.- Inserta un deslizador **n** que tome valores enteros entre 1 y 100. Determinará las raíces n-ésimas que se desean calcular

2.- Inserta un deslizador **a** que tome valores enteros entre -1000 y 1000. Determinará el valor de **la parte real** del complejo.

3.- Inserta un deslizador b que tome valores enteros entre -1000 y 1000. Determinará el valor de la **parte imaginaria** del complejo.

4.- Introduce la casilla de entrada **c** que tome valores para **a**. Para ello: **CasillaEntrada(a)**

5.- Introduce la casilla de entrada **d** que tome valores para **b**. Para ello: **CasillaEntrada(b)**

6.- Introduce en la barra de comandos **h=a + b í**

7.- Introduce en la barra de comandos **modulo=Longitud(h)**

8.- Introduce en la barra de comandos **m=modulo^(1/n)**

9.- Introduce en la barra de comandos
raices=Secuencia((m*cos((α+2*k*π)/n), m*sen((α+2*k*π)/n), k, 0, n - 1, 1)

10.-Crea un texto en el que aparezca "raices"

11.- Retócalo gráficamente para que quede con colores adecuados.

Guarda el archivo generado con el nombre **complejo.ggb**

ACTIVIDAD NÚMERO 13

Título de la actividad:	Lista de números primos

Enunciado de la actividad

El propósito de esta actividad es obtener los números primos existentes entre dos números dados

Pasos que se seguirán con GeoGebra

Para hacerlo, sigue los siguientes puntos paso a paso:

1.- Inserta un deslizador **n** que tome valores enteros entre 3 y 5000.

2.- Inserta un deslizador **m** que tome valores enteros entre n+1 y 5000.

3.- Crea la siguiente secuencia **r=Secuencia(PrimoAnterior(t), t, n, m, 1)**

4.- Introduce en la barra de comandos **p=Único(r)**

5.- Ahora vamos a prepararlo para que en la pantalla aparezca el mayor número de primos posible

6.- Introduce en la barra de comandos **l=Longitud(p)**

7.- Introduce en la barra de comandos **l1=l/4 - parteFraccionaria(l/4)**

8.- Introduce en la barra de comandos **l2=2*l/4 - parteFraccionaria(2*l/4)**

9.- Introduce en la barra de comandos **l3=3*l/4 - parteFraccionaria(3*l/4)**

10.-Introduce en la barra de comandos **r1=Secuencia(Elemento(p, q), q, 1, l1, 1)**

11.-Introduce en la barra de comandos **r2=Secuencia(Elemento(p, q), q, l1+1, l2, 1)**

12.-Introduce en la barra de comandos **r3=Secuencia(Elemento(p, q), q, l2+1, l3, 1)**

13.-Introduce en la barra de comandos **r3=Secuencia(Elemento(p, q), q, l3+1, l, 1)**

14.- Coloca en la pantalla el texto "Existen **[l]** primos entre **[n]** y **[m]** que son:"

15.- Debajo coloca las listas **r1**, **r2**, **r3** y **r4**

16.- Retócalo gráficamente para que quede con colores adecuados.

Guarda el archivo generado con el nombre **lista_primos.ggb**

ACTIVIDAD NÚMERO 14 (PROFUNDIZACIÓN)	
Título de la actividad:	Visualizar los múltiplos comunes

Enunciado de la actividad

Dados dos números n y m, representa visualmente los múltiplos comunes y obtén en esa representación el mínimo común múltiplo.



Pasos que se seguirán con GeoGebra

Para hacerlo, sigue los siguientes puntos paso a paso:

1.- Oculta el **Eje Y** y modifica la escala del eje X para que abarque más números. Para ello, clic derecho sobre la Vista gráfica, desmarca la casilla Muestra de la ficha Eje Y.
En la ficha Básico, modifica los valores mínimo y máximo del Eje X. coloca de 0 a 300

2.- Inserta dos deslizadores para controlar los valores de cada uno de los dos naturales (**m y n**) cuyos múltiplos queremos visualizar:

3.- Inserta otros dos deslizadores que controlarán el número de múltiplos visibles de m y de n. Llama a estos deslizadores, que variarán entre **1 y 50**, **cantidad_m** y **cantidad_n**

4.- Calculemos las dos listas de múltiplos. Para ello teclea en el Campo de Entradas: **lista1= Secuencia[m*i,i,1,cantidad_m]** y luego comprueba cómo al variar los valores de m y cantidad_m, mediante los deslizadores correspondientes, se actualizan (en la Vista algebraica) los múltiplos de **m**.

5.- Repitamos el proceso para obtener una segunda **lista2= Secuencia[n*i,i,1,cantidad_n]** de los primeros multiplos de **n**.

6.- Creamos una tercera lista con los múltiplos comunes. Para ello teclea en el Campo de Entradas: **lista3= Intersección[lista1,lista2]**

7.- Tras comprobar cómo al cambiar los valores de m y n, se actualizan los valores de las listas creadas, vamos a visualizarlas sobre la pantalla gráfica en forma de puntos sobre el eje de abcisas. Para ello visualizaremos esas listas de números como puntos sobre el eje horizontal y personalizaremos sus estilos para diferenciarlos claramente.

8.- Teclea en el Campo de Entradas: **lista4= Secuencia[(Elemento[lista1,i],0),i,1,cantidad_m]**. Luego edita sus propiedades aumentando su tamaño y eligiendo un estilo diferente.

9.- Repite el proceso para la lista de múltiplos de **n**:
lista5=Secuencia[(Elemento[lista2,k],0),k,1,cantidad_n]. Luego edita sus propiedades aumentando su tamaño y eligiendo un estilo diferente.

10.- Señalaremos los múltiplos comunes mediante unas flechitas hacia abajo (en realidad puntos con ese estilo). Para ello teclea en el Campo de Entradas:
comunes=Secuencia[(Elemento[lista3,i],0.5),i,1,Longitud[lista3]]. Luego edita sus propiedades para que esos puntos aparezcan como flechitas para abajo.

11.- En la ventana de entrada colocamos **G=Mínimo(lista3)**

12.- Representaremos ahora el mínimo común múltiplo mediante un punto bajo el eje de abscisas. Para ello teclea en el Campo de Entradas: **MCM=(G,-0.5)**. Luego edita sus propiedades para que, además de visualizarlo como un triangulito-flecha hacia arriba, oculta el nombre.

13.- Para dejarlo bonito, ponemos un punto oculto **B=(G, -0.9)**

14.- Apoyado en ese punto, introducimos el texto **MCM=[G]**

15.- Retócalo gráficamente para que quede con colores adecuados.

Guarda el archivo generado con el nombre **multiplos.ggb**

La siguiente actividad está tomada de un taller del grupo Xeodin.

ACTIVIDAD NÚMERO 15	
Título de la actividad:	Suma de una progresión aritmética
Enunciado de la actividad	

Con esta actividad se pretende visualizar la suma de los términos de una progresión aritmética. Esa progresión se va a presentar como una serie de rectángulos de altura constante y amplitud correspondiente a la progresión. Esos rectángulos formarán como una escalera, que al girarla veremos cómo se forma un rectángulo que será el doble de la suma.



Pasos que se seguirán con GeoGebra

Comenzamos creando tres deslizadores: a_1 y d, ambos entre 1 y 5, y un tercer deslizador n entre 1 y 10. Todos con incremento la unidad, para trabajar con números naturales. Los colocamos en la parte superior de la ventana, alineados.

Situamos un punto A en la parte superior derecha de la ventana, debajo de los deslizadores. Se incluyen ahora los puntos ***B=A+(0,-1)*** y ***C=B+(1,0)***.

Estos tres puntos nos darán el vértice para construir los rectángulos. Debes ocultarlos pues sólo nos servirán de referencia.

Debemos construir ahora la lista de rectángulos a partir de esos puntos. Para ello debes copiar la siguiente orden.

L1 = Secuencia(Polígono(A + (i - 1) (B - A), A + i (B - A), A + i (B - A) + (a_1 + (i - 1) d) (C - B), A + (i - 1) (B - A) + (a_1 + (i - 1) d) (C - B)), i, 1, n)

Con esto ya tenemos dibujada la escalera de los términos de la progresión. Vamos a ampliarla colocándole junto a cada rectángulo el valor de la progresión. Para ello, introducimos dos nuevas órdenes:

prog = Secuencia(a_1 + (t - 1) d, t, 1, n)

text = Secuencia(Texto(Elemento(prog, t), PuntoMedio(A, B) - (1, 0.25) - (t - 1) (0, 1)), t, 1, n)

Ahora viene la parte visual para observar la suma de los términos de la progresión. En primer lugar necesitamos su término final que lo conseguimos con la orden:

a_n = a_1 + (n-1) d

Construimos ahora un nuevo número que será el opuesto de A en la construcción escalonada. Igual que los anteriores lo ocultamos.

D=A+(a_1+a_n,-n)

Y el punto que nos va a servir para girar la estructura será:

M=(A+D)/2

Se construye un nuevo deslizador α, de tipo ángulo variando desde 0º a 180º. En propiedades, le añadimos como rótulo *suma* y hacemos visible en la etiqueta el rótulo.

Rotamos los puntos A, B y C respecto a M un ángulo igual a α. Todos los puntos nuevos deben ser ocultados.

La siguiente orden será la misma que hicimos antes para crear los rectángulos, pero particularizada ahora para los nuevos puntos.

L2 = Secuencia(Polígono(A' + (i - 1) (B' - A'), A' + i (B' - A'), A' + i (B' - A') + (a_1+ (i - 1) d) (C' - B'), A' + (i - 1) (B' - A') + (a_1 + (i - 1) d) (C' - B')), i, 1, n)

Para que quede más visual, vamos a utilizar colores dinámicos para la lista que movemos. Para ello elegimos la lista L2 y en el apartado de propiedades, dentro de la pestaña Avanzado colocamos los siguientes valores:



Si ahora movemos el deslizador suma podemos ver como se visualiza la suma doble de los términos de la progresión.

Bastaría completar la construcción añadiendo una serie de textos que den más información sobre los valores manejados en la presentación.

Guarda el archivo generado con el nombre **progresión_aritmetica.ggb**

A veces, la orden secuencia no nos permite obtener fácilmente la sucesión que queremos. Esto ocurre, por ejemplo, con las sucesiones recurrentes. No es posible conseguir una sucesión, como la de Fibonacci, si no conocemos la expresión del término general.

Vamos a ver que es muy fácil trabajar con la hoja de cálculo de GeoGebra y conseguir una lista de una serie de elementos. Para ello, una vez que tengamos seleccionados los elementos basta pulsar en el icono de la barra de herramientas y seleccionar *Lista*.





Al construir la lista podemos escoger que sea dependiente o libre. Si elegimos la primera opción, cualquier cambio que hagamos en la hoja de cálculo se actualizará en la lista.

Para la siguiente actividad vamos a necesitar tomar los primeros términos de una lista, para ello usaremos la orden:

Primero(lista, número)

Con ella se toman los primeros elementos, hasta el de lugar n, de la lista citada.

ACTIVIDAD NÚMERO 16	
Título de la actividad:	Sucesión de Fibonacci.

Enunciado de la actividad

Vamos a construir la sucesión de Fibonacci y a comprobar, que a medida que avanzamos en la sucesión, el cociente entre dos términos consecutivos se acerca al número de oro.



Pasos que se seguirán con GeoGebra

Abrimos la hoja de cálculo y escribimos 1 en las casillas A1 y A2. En la casilla A3 escribimos la expresión ***=A1+A2***. Una vez introducida, seleccionamos la celda y la arrastramos hasta la casilla A50.

Pulsando en el icono de lista, tras haber seleccionado la columna A, convertimos el contenido de la columna en una lista que llamaremos ***fibo***.

En la ventana gráfica creamos un deslizador ***n*** con valores desde 5 a 50 e incremento de 1. Creamos una lista de texto, para presentar en la ventana la sucesión. Utilizamos el comando:

texto1 = Texto(Primero(fibo, n))

Creamos un nuevo texto para que aparezca el nombre de la sucesión y otro para el cálculo del cociente. Creamos dos nuevos objetos

tn=Elemento(fibo,n)

tn_1=Elemento(fibo,n-1)

Para el cálculo, escogemos en el texto la opción de *fórmula Latex* y escribimos la orden:

\frac{f_{ n }}{f_{ n-1 }}= tn/tn_1

Hay que recordar que n, n-1, tn y tn_1 hay que elegirlos entre los objetos ya definidos.

Guarda el archivo generado con el nombre **fibonacci2.ggb**

Como ampliación del ejercicio anterior vamos a proponer estudiar la Sucesión de Lucas, que es también una sucesión recurrente en la que el término cero vale 2, el primero 1, y los restantes se obtienen como en Fibonacci, sumando los dos anteriores.

ACTIVIDAD NÚMERO 17 DE AMPLIACIÓN	
Título de la actividad:	Sucesiones de Fibonacci y Lucas
Enunciado de la actividad	

Construye las sucesiones de Fibonacci y de Lucas, sabiendo que:

$$f_0 = 0 \ ; \ f_1 = 1 \ ; \ f_n = f_{n-1} + f_{n-2} \ para \ n \geq 2$$
$$l_0 = 2 \ ; \ l_1 = 1 \ ; \ l_n = l_{n-1} + l_{n-2} \ para \ n \geq 2$$

Una vez construidas las dos sucesiones, comprueba que se verifican las siguientes igualdades:

a) $l_n = f_{n-1} + f_{n+1} \ para \ n \geq 1$
b) $l_n = f_{n+2} + f_{n-2} \ para \ n \geq 2$
c) $f_{2n} = f_n \cdot l_n \ para \ n \geq 1$

A la hora de manejar las sucesiones, ten presente que el primer término será el correspondiente subcero de la sucesión. Si no tienes esto en cuenta, tendrás problemas con el apartado c. Otra opción es considerar que Fibonacci comienza con 1, 1 y que Lucas comienza con 1, 3.



ACTIVIDAD NÚMERO 18	
Título de la actividad:	Lanzamiento de un dado

Enunciado de la actividad

Vamos a simular el lanzamiento de un dado y a estudiar las frecuencias relativas de los distintos resultados, mediante un gráfico de barras. Así visualizaremos la Ley de los Grandes Números.



Pasos que se seguirán con GeoGebra

Creamos un deslizador n con valores desde 10 hasta 1000 y con incremento 10.

Generamos una lista llamada ***tirada*** con mil números elegidos aleatoriamente entre 1 y 6.

Generamos dos nuevas listas con las frecuencias absolutas y relativas de los primeros n números de la lista:

FreAbs=Frecuencia(Primero(tirada,n))

FreRel=Secuencia(Elemento(FreAbs,t)/n,t,1,6)

El siguiente paso es crear el diagrama de barras, con la orden:

Barras({1,2,3,4,5,6},FreRel)

Para completar la construcción bastaría señalar con un segmento la frecuencia relativa teórica, que sabemos que en un dado de seis caras es 1/6.

Segmento((0,1/6),(7,1/6))

Basta aumentar el deslizador para ver como las frecuencias se van acercando al valor teórico.

Guarda el archivo generado con el nombre **fibonacci_lucas.ggb**

ACTIVIDAD NÚMERO 19	
Título de la actividad:	Fractal de Sierpinski
Enunciado de la actividad	

Utilizando listas es muy fácil generar un fractal. Vamos a hacerlo en la ventana 3D generando la pirámide de Sierpinski.



Pasos que se seguirán con GeoGebra

Comenzamos dibujando un tetraedro, para ello vamos a dibujar dos puntos cualesquiera y el triángulo equilátero sobre ellos. Para después construir el primer tetraedro.

A=(-5,-5,-4)

B=(-4,-5,-4)

Polígono(A,B,3,EjeZ)

T1=Tetraedro(A,B,C) (suponemos que C es el tercer ángulo creado por la orden anterior)

El siguiente paso es construir una lista de vectores que nos van a indicar las direcciones en que vamos a trasladar las listas de tetraedros.

LisVec={Vector(A,A), Vector(A,B), Vector(A,C), Vector(A,D)}

Aunque el primer vector nos va a repetir la lista anterior, lo ponemos para no tener que añadir una iteración del fractal a la siguiente. Es posible hacerlo, pero crea objetos más grandes y al final bloquea la construcción.

A partir de aquí, es repetir la orden siguiente las veces que queramos:

T2=Secuencia(Traslada(T1,Elemento(LisVec,i)),i,1,4)

T3=Secuencia(Traslada(T2,2 Elemento(LisVec,i)),i,1,4)

T4=Secuencia(Traslada(T3,4 Elemento(LisVec,i)),i,1,4)

Y así sucesivamente, sólo hay que ir multiplicando, en cada iteración, los vectores desplazamiento por la siguiente potencia de 2: 8, 16, …

Guarda el archivo generado con el nombre **sierpinski.ggb**

En la siguiente actividad vamos a visualizar, en la ventana 3D, los puntos correspondientes a las tablas de multiplicar. Sería posible hacerlo directamente con listas, pero es más fácil hacerlo directamente con la hoja de cálculo.

ACTIVIDAD NÚMERO 20

Título de la actividad:	Tabla de multiplicar.

Enunciado de la actividad

Dibujar los puntos de la forma (a, b, a·b) con a y b cambiando desde -5 hasta 5. Unir los puntos extremos mediante líneas para visualizar la multiplicación gráficamente.



Pasos que se seguirán con GeoGebra

Antes de comenzar es conveniente seleccionar en Opciones el Etiquetado y seleccionar que no se ponga a ningún objeto nuevo.

Abre la ventana 3D y la hoja de cálculo. En la casilla B1 introduce el número 5 y en C1 el 4. Selecciona los dos y arrastra hasta L. Obtendrás los valores desde 5 hasta -5. Haz lo mismo en la primera columna, escribiendo en A2 5, en A3 4 y arrastrando hasta la fila 12.

En la casilla B2 introduce la expresión ***=(A2, B1, A2 B1)*** y arrastra la celda hasta L2. Haz lo en las restantes filas, fijando en cada caso la casilla correspondiente a la columna con la fila en que te encuentres. Es decir, en B3 debes escribir ***=(A3, B1, A3 B1)*** y así hasta la fila 12.

En la casilla B13 se incluye la orden ***=Segmento(B2, B12)*** y se arrastra la orden hasta la casilla L13. De manera análoga, en M2 se incluye la orden ***=Segmento(B2, L2)***, y luego se arrastra la celda hasta la casilla M12.

Nos aparecerá en la ventana 3D una superficie reglada donde los puntos son las representaciones de los productos de los números desde -5 hasta 5.

Si queremos completar la actividad con la vista de la superficie, basta incluir, en la ventana 3D, la orden

Superficie(a, b, a b, a, -7, 7, b, -7, 7)

Así obtendremos, sobre las líneas anteriores, la superficie parabólica.

Guarda el archivo generado con el nombre **multiplicar.ggb**

ACTIVIDAD NÚMERO 20 (PROFUNDIZACIÓN)	
Título de la actividad:	Probabilidad en triángulo.
Enunciado de la actividad	
	En un triángulo cualquiera seleccionamos un punto al azar en cada uno de sus lados. ¿Qué probabilidad existe de que el área del triángulo dibujado sea inferior 1/i parte del área del triángulo inicial? Realiza una construcción con GeoGebra en la que se aproxime esa probabilidad. Selecciona i entre 1 y 10
Pasos que se seguirán con GeoGebra	

Ahora toca pensar….

1.- Crea un deslizador **n** que tome valores entre 1 y 2000. Indicará el número de veces que realizaremos el experimento de coger un triángulo que tenga cada uno de sus vértices en cada lado del triángulo inicial.

2.- Crea un deslizador **i** que tome valores enteros entre 1 y 10

3.- Crea la secuencia **a=Secuencia((random() t) / t, t, 1, n)**. Es una lista de **n** números aleatorios entre 0 y 1.

4.- Crea la secuencia **b=Secuencia((random() t) / t, t, 1, n)**. Es una lista de **n** números aleatorios entre 0 y 1.

5.- Crea la secuencia **c=Secuencia((random() t) / t, t, 1, n)**. Es una lista de **n** números aleatorios entre 0 y 1.

6.- Marca en la ventana gráfica tres puntos **A**, **B** y **C**. Serán los vértices del triángulo original

7.- Introduce en la ventana de comandos **ab=Vector(A, B)**

8.- Introduce en la ventana de comandos **ac=Vector(A, C)**

9.- Introduce en la ventana de comandos **bc=Vector(B, C)**

10.- Introduce en la ventana de comandos **L=Secuencia(B + Elemento(c, s) bc, s, 1, n)**

11- Introduce en la ventana de comandos **M=Secuencia(A + Elemento(a, k) ab, k, 1, n)**

12.- Introduce en la ventana de comandos **N=Secuencia(A + Elemento(b, h) ac, h, 1, n)**

13.- Introduce en la ventana de comandos **p=Polígono(A, B, C).** Representa el triángulo principal.

14.- Introduce en la ventana de comandos **q=Secuencia(Polígono(Elemento(M, w), Elemento(N, w), Elemento(L, w)), w, 1, n)**. Dibuja cada uno de los triángulos que tienen un vértice aleatorio en cada lado del anterior.

15.- Introduce en la ventana de comandos **Ar=Secuencia((1 / i p - Elemento(q, o)) / abs(1 / i p - Elemento(q, o)), o, 1, n)** Con esto obtenemos el valor 1 si el área del triángulo aleatorio es menor que 1/i del área del triángulo principal y obtenemos el valor -1 si el área del triángulo aleatorio es mayor que **1/i** del área del triángulo principal

16.- Introduce en la ventana de comandos **Br=Secuencia(Máximo(0, Elemento(Ar, j)), j, 1, n)** Con esto obtenemos el valor 1 si el área del triángulo aleatorio es menor que 1/i del área del triángulo principal y obtenemos el valor 0 si el área del triángulo aleatorio es mayor que 1/i del área del triángulo principal.

17.- Introduce en la ventana de comandos **Tot=Suma(Br)** esto nos suma todos los valores de la lista generada en el punto anterior, por lo que obtenemos el número de triángulos aletorios que cumplen que el área del triángulo aleatorio es menor que 1/i del área del triángulo principal.

18.- Introduce en la ventana de comandos **d=Tot/n**

19.- En la ventana gráfica introduce el texto "Dado un triángulo cualquiera, si se selecciona un punto al azar de cada uno de sus lados, la probabilidad de que el área del triángulo delimitado por estos tres vértices sea inferior a 1**/[i]** del área del triángulo inicial es **[d]**. Utilizamos **[n]** iteraciones

20.- Retócalo gráficamente para colocarle los colores y posiciones que consideres necesarios. Quita la visibilidad de ejes y cuadrícula.

Guarda el archivo generado con el nombre **triangulos.ggb**

Ampliación de la construcción con GeoGebra

En un cuadrilátero cualquiera seleccionamos un punto al azar en cada uno de sus lados y dibujamos un nuevo cuadrilátero que tenga por vértices esos puntos. ¿Qué probabilidad existe de que el área del cuadrilátero dibujado sea inferior 1/i parte del área del cuadrilátero inicial?

Realiza una construcción con GeoGebra en la que se aproxime esa probabilidad. Selecciona i entre 1 y 10

Guarda el archivo generado con el nombre **cuadrilátero.ggb**